
ABSTRACT 




The Base SAS® Output Delivery       System (ODS) makes it easy to generate       reports for viewing on desktops.  What about mobile devices?  If you       need on-the-go reports, then the new SAS 9.4 ODS EPUB destination is       the ticket.  With ODS EPUB, you can generate your reports as e-books       that you can read with iBooks on the iPad, or, you can write an e-book       from scratch if you like.

This paper provides an introduction to writing e-books with ODS EPUB.        Please bring your iPad, iPhone, or iPod to the presentation so that       you can download and read the examples.







INTRODUCTION 




If you use ODS, then you are accustomed to having ready access to SAS       reports, whether in your home or at work with your desktop computer,       or anywhere with your laptop computer.  ODS provides various output       formats to fit the reporting task at hand.  ODS HTML is well-suited       to report development and debugging, especially in the Display Manager       System (DMS) environment.  ODS RTF is ideal for including reports in       word processing documents. ODS PDF works best for printing and       distributing reports.  These scenarios assume a typical home or work       computing configuration, which includes a monitor, keyboard, and       pointing device connected to one or more computers running SAS.

If you are like many people, you have one or more mobile devices       that you use at home, work, and wherever you go.  Naturally you       wish to consume your SAS reports on your tablet or smartphone.        However, the aforementioned output formats were not designed for       mobile computing, and each has its drawback.  HTML requires a web       browser connected to the Internet, and is not fully portable.        RTF cannot be viewed natively on mobile devices.  PDF is formatted       to the printed page, and does not scale down well to mobile displays.        How can you create SAS reports that are portable and viewable on       mobile devices of varying form factors?

The SAS 9.4 ODS EPUB destination enables you to create SAS reports       as e-books that look great and travel well on your smartphone or       tablet.  ODS EPUB is a part of Base SAS, and runs everywhere that       Base SAS runs.  Note that on z/OS, ODS EPUB requires the HFS file       system.

The ODS EPUB destination is designed to generate e-books that work       optimally with the Apple iBooks e-book reader on iPad, iPhone, and       iPod.  Results might vary with other e-book readers.

After reading the paper or viewing the presentation, any user should       be able to view the example reports on an iPad, iPhone, or iPod.        Novice users should be able to run the example SAS jobs to generate       and view mobile reports.  Experienced users should be able to create       mobile reports, or to write an e-book from scratch.  The desired result       is that you will "cut the cord", take your SAS reports on the road,       and never look back!







GET STARTED 




Your first order of business is to learn how to specify ODS EPUB       statements to generate a SAS report as an e-book.  After that you       will learn several techniques for loading EPUB e-books into the       iBooks e-book reader.  At that point you will know enough to go       mobile with ODS EPUB!





MAKE A SANDWICH


The SAS 9.4 ODS EPUB statement creates output in the EPUB format.        EPUB is a free and open e-book standard by the International Digital       Publishing Forum (IDPF).  EPUB files have the extension        .epub.  ODS       EPUB e-books have no Digital Rights Management (DRM) restrictions and       can be freely copied and shared.  Numerous e-book readers support EPUB       e-books.  Popular free EPUB e-book readers include Adobe Digital       Editions, Calibre, EPUBReader, Kobo, and Stanza.  The recommended       e-book reader for ODS EPUB e-books is Apple iBooks on iPad, iPhone,       and iPod.







The easiest way to get started with the ODS EPUB statement is to try       it with a simple example that produces familiar output.  Consider       this SAS code:






[image: images/epubimage3.png]


Show preformatted output
Example 1







Like other ODS destinations, the ODS EPUB destination supports the      "ODS sandwich" programming idiom.  You sandwich your procedure       and DATA steps in ODS EPUB statements.  The beginning ODS EPUB       statement opens an ODS EPUB destination and prepares it to process       the procedure and DATA step output.  The ending ODS EPUB statement       closes the destination, which terminates the generation of output       to the specified file.  An open ODS EPUB destination remains open       until it is explicitly closed or the SAS session ends.

The beginning ODS EPUB statement in the above example specifies       three options.  FILE= specifies the output filename.  If you specify       no name, ODS EPUB will supply a default name (typically        sasepub.epub).        TITLE= specifies the title of your e-book.  If you specify no TITLE=       title, ODS EPUB will supply a default title of "SAS Output" or its       localized equivalent.

If you want to specify one or more authors for your e-book, use the       CREATOR suboption of the ODS EPUB statement OPTIONS option.  Delimit       multiple authors with commas.  If you do not specify an author, your       e-book will not have one.  You probably want your e-book to have an       author!

In the example, the content of the "ODS sandwich" is a PROC FREQ       step that requests a frequency cross-tab and plot.  Figure 1 shows       how the SAS 9.4 DMS internal browser renders the default HTML that       ODS generates for the report.






[image: A screenshoot of SAS Results Viewer, displaying the HTML output of the sample SAS code from 5 paragraphs earlier.]

Figure 1. Frequency Output in SAS 9.4 DMS





The SAS Log confirms that the ODS EPUB destination created the file       , which is the e-book  Cars'       Origin by Type:

Show preformatted output
Output 1. SAS Log for Frequency Report







SAS does not configure a default EPUB e-book reader, so DMS does not       display the contents of  Cars' Origin by Type.  Having       viewed the DMS HTML output, you can imagine what the EPUB output       might look like, but you want to be able to read it on your mobile       device.  To do that, you need to load  cars_freq.epub       into an EPUB       e-book reader–but how, and with what e-book reader?





READ WITH IBOOKS


ODS EPUB e-books are designed to work best with iBooks.  iBooks is a       free e-book reader app developed by Apple for iOS devices.  iBooks is       not preinstalled on your device–like most apps, you       install it from the App Store.







iBooks enables you to download and read e-books from Apple's       iBookstore.  You also can sideload EPUB e-books into iBooks from       whatever source you specify.  Sideloading is the method required to       load e-books created with ODS EPUB into iBooks.





YOU'VE GOT MAIL


As you might expect for an Apple mobile device, you need an app for       sideloading e-books.  As you might also expect, there is more than one       app that can do the job.  Which one should you use?  For getting       started, go with the app that you probably already know and       use–Apple's built-in Mail app.  Simply attach the EPUB       file to an e-mail message that you send to yourself.  Then access the       message with the Mail app.  Tapping the e-mail attachment icon will       bring up a list of apps for viewing the attachment.  Figure 2 shows       what this looks like for EPUB file attachment        cars_freq.epub.









[image: Screenshot of iOS mail client, with an email attachment entitled 'cars_freq.epub'. A pop up in the screen shot indicated over 9 options for opening an .epub file. The first one is 'Open in iBooks']

Figure 2. Apps for Viewing an EPUB E-book (Mail)





Selecting the list icon for iBooks opens the e-book in iBooks, which       automatically adds the e-book to its library. Figure 3 shows how       iBooks displays the frequency output on a third-generation iPad.






[image: A screenshot of iBooks on an iPad rendering cars_freq.epub. The generated table and bar charts all fit on one page.]

Figure 3. Frequency Output in iBooks





iBooks is designed to simulate the experience of reading a physical       book.  Using gestures, you "turn pages" left or right.  Page 1 is a       right page, just like with a physical book.  (Apple is famous for       applying such skeuomorphic design principles to its products.)  The       size of the page limits the space available for tables and graphic       displays.  iBooks provides a way to maximize those tables and graphic       displays, taking advantage of the entire screen.

If you want to inspect a table or graphics display more closely, then       double-tap it on the screen.  iBooks will launch into a full-screen       mode that allows you to pan and zoom the table or graphics display.        To pan and zoom, simply pinch the output on the screen.  Figure 4       shows the frequency cross-tab in iBooks pan-and-zoom mode.






[image: A screenshot of the PROC FREQ results table, in full-screen mode, panned, and zoomed so that only the core table is viewable.]

Figure 4. Zoomed Frequency Cross-tab in iBooks





Another way to maximize your screen real estate is to put iBooks in       scroll mode.  You can set scroll mode via the Fonts user interface.        Figure 5 displays the frequency report in scroll mode.






[image: A screenshot of iBooks, with the fonts user interface pop up visible and the Scroll option checked.]

Figure 5. Scrolled Frequency Report in iBooks





Every ODS EPUB e-book has a table of contents, which will be familiar       to ODS users.  Every iBooks e-book also has a cover.  To see the table       of contents and cover, first find the table of contents icon on the       upper left of the screen, next to the Library button.  If you tap that       icon, iBooks will pause reading and display the e-book's cover and       table of contents.  See Figure 6.






[image: A screenshot of the Table of Contents view of cars_feq.epub]

Figure 6. Cover and Table of Contents in iBooks





Selecting an entry in the table of contents takes you to the associated       page in the e-book.

You can customize the table of contents and cover.  You will learn how       to do that later in the discussion.

Before digging deeper into iBooks' functionality, you need to learn       more about getting your ODS EPUB e-books into iBooks.  Sideloading with       the Mail app is just one way to do it.  What are some other ways?





THERE'S ANOTHER APP FOR THAT


Sideloading an EPUB e-book into iBooks with the Mail app can have       drawbacks.  E-mail servers can impose restrictions on attachments that       make it difficult to e-mail an EPUB e-book.  If the e-book contains       many graphics displays, then its size might exceed the server-defined       limit.   Some e-mail servers reject ZIP file attachments, which means       that they reject EPUB file attachments, because an EPUB file is a ZIP       file.  Some e-mail servers reject attachments of any kind.  The Mail       app cannot help you if your EPUB file attachment has been stripped       from your e-mail!







Apple builds into its mobile devices another app that can sideload       EPUB e-books into iBooks without the potential pitfalls of e-mail       delivery: Safari.  Safari is Apple's web browser.  If you place your       EPUB files where your web server can find them, then Safari can find       them also.

Typically, web servers are set up so that users can make files       web-accessible by placing them in a  public_html       directory.  If so, placing the  cars_freq.epub file       in a  public_html directory will make it visible to       Safari.  Your web server might  be set up this way.  Consult your web       administrator if you do not know.

However you choose to do it, once you have made your EPUB file       web-accessible, you can browse it with Safari.  Figure 7 shows Safari       listing the files in a directory.   cars_freq.epub       is one of the files.






[image: A screenshot of a web page, with cars_feq.epub as a hyperlink.]

Figure 7. Directory Listing in Safari





If you select  cars_freq.epub from the list, Safari       will prompt you to specify the app to open it.  See Figure 8.






[image: A screenshot of Safari on iPad loading cars_freq.epub. Safari displays the file name, the size of the file (36kb), and two buttons: 'Open in...' and 'Open in iBooks']

Figure 8. Apps for Viewing an EPUB E-book (Safari)





Open the e-book in iBooks, and you are ready to start reading!





ROCK YOUR DROPBOX


Safari suffices for sideloading EPUB e-books into iBooks, provided       that you have access to a web server.  What if you do not?  You can       fall back to the Mail app, but you might run into issues with your       e-mail server, as was discussed earlier.







What if you could store your e-books in the cloud and access them       from your smartphone or tablet?  That would avoid any e-mail or web       server issues.  Cloud-based storage is plentiful and       cheap–even free perhaps, depending on how much storage       you consume.

Cloud storage providers are numerous, and include Google Drive,       Microsoft SkyDrive, and Apple iCloud.  One of the most popular       providers of consumer-based cloud storage is Dropbox       (by Dropbox, Inc.).  Dropbox stores customer files in a synchronized,       version-controlled virtual file system.  Dropbox provides free client       software for reading and writing the files.  The client software       runs on Windows, Mac, and Linux desktops, as well as Android, iOS,       and Blackberry devices.  The first 2 gigabytes of cloud storage are       free, and you can acquire additional storage for free, if you do not       mind volunteering to help Dropbox solicit new customers!

Dropbox's most important feature is that it is easy to use.  When you       install Dropbox on a Windows machine, it integrates the cloud file       system seamlessly into the operating system.  See Figure 9.






[image: A screenshot of Windows Explorer displaying 10 EPUB files in Dropbox/BooksByODS]

Figure 9. Dropbox Directory Listing in Windows





If you place an EPUB file, or any file, in your Dropbox directory,       then Dropbox will automatically synchronize it to the cloud file       system.  Because the Dropbox directory is an actual Windows directory,       it is simple to integrate Dropbox with SAS in general and the ODS EPUB       destination in particular.  Consider the following revision to the       frequency report SAS code:

Show preformatted output
Example 2







When this code is run in SAS 9.4 on the author's Windows PC, the      ODS EPUB destination writes the  cars_freq.epub file      directly into the  BooksByODS subdirectory in the      Dropbox directory.  Dropbox synchronizes the EPUB file to the     cloud file system, after which the e-book is accessible to any      computer linked to the Dropbox account, including the author's iPad.

Invoking the iPad Dropbox app tells the tale.  Navigating to the       BooksByODS subdirectory and selecting       cars_freq.epub from the directory listing downloads      the file from the Dropbox cloud storage onto the device.  After that,      tapping the icon in the upper right corner of the screen launches the      App Selector dialog box.  See Figure 10.






[image: A screenshot of the Dropbox iOS app. Dropbox's response to 'cars_freq.epub' is 'Unable to view file'. The 'Download' pop up reveals several apps that can open an EPUB file, the first option being 'Ope]

Figure 10. Apps for Viewing an EPUB E-book (Dropbox)





Open in iBooks, and you are done.  That was easy!

Beyond its ease of use, Dropbox enables you to automate the generation      and distribution of SAS reports.  You can share Dropbox cloud      directories with coworkers, friends, and relatives.  When you create      or update an EPUB e-book, everyone linked to your Dropbox directory      can access the e-book.

Now that you know how to create ODS EPUB e-books, and how to sideload      them into iBooks, you know enough to go mobile.  You can get your      traditional SAS reports out of your office and onto the devices you      carry around with you daily.  However, doing only that provides you      only partial benefits.  The ODS EPUB destination is designed to enable      you to write books, and not just reports. To take advantage of that      capability, you need to go beyond understanding how to generate and      distribute e-books with ODS EPUB.  You need to learn how to use ODS      EPUB to write rich text, create chapters, and customize the table of      contents.  Doing so will take you into new territory as a SAS user.       Read on!






title;                                                                          
ods epub file="cars_freq.epub" title="Cars' Origin by Type"                     
                               options(creator="David W. Kelley");              
proc freq data=sashelp.cars;                                                    
tables origin*type/plot=freq;                                                   
run;                                                                            
ods epub close;                                                                 




NOTE: Writing HTML Body file: sashtml.htm                                       
NOTE: There were 428 observations read from the data set SASHELP.CARS.          
NOTE: PROCEDURE FREQ used (Total process time):                                 
      real time           3.61 seconds                                          
      cpu time            0.35 seconds                                          
                                                                                
                                                                                
8    ods epub close;                                                            
NOTE: Writing package .\cars_freq.epub.                                         




filename ebook "c:\users\sasdck\dropbox\booksbyods\cars_freq.epub";             
title;                                                                          
ods epub file=ebook title="Cars' Origin by Type"                                
                    options(creator="David W. Kelley");                         
proc freq data=sashelp.cars;                                                    
tables origin*type/plot=freq;                                                   
run;                                                                            
ods epub close;                                                                 





BOOK IT 




Up to this point, the discussion has been about how to generate a SAS       report as an EPUB e-book and read it with the iBooks e-book reader.        Knowing how to do that will get you only so far.  Distributing a       traditional SAS report as an e-book does not make the report a book.        A book has distinctive features that you do not find in a report. The       most obvious difference is the structure of text.  Books organize       words into sentences, sentences into paragraphs, and paragraphs into       sections and chapters.  SAS has never allowed customers to do       that–until now.





ENRICH TEXT


SAS 9.4 ODS changes the game.  In SAS 9.4, text joins tables and       graphic displays as a first-class ODS reporting element.  In the       parlance of ODS, text becomes an  output object, with       all of the attendant behaviors familiar to ODS users.  With this new       capability, you can write whatever you like–an essay,       a shopping list, or an instruction manual–with SAS.        You can then format it with one or more destinations, including ODS       PDF and ODS POWERPOINT.







Nowhere is this new rich text feature demonstrated to better effect       than with the ODS EPUB destination.  Consider this example:






[image: images/epubimage3.png]


Show preformatted output
Example 3







When you run the example in SAS 9.4, ODS EPUB generates the EPUB       e-book  Visit 50 States.  Figure 11 shows how iBooks       displays the e-book.






[image: A screenshot of iBooks displaying the first version of visit_50_states.epub. The title is 'Visit 50 States' and the author is 'David W. Kelley'. The book is 1 page long.]

Figure 11. Rich Text in iBooks





That does not look like the SAS output you know and love!  There is       not a table or graphics display in sight.  It is just       text–specifically, two paragraphs and a list.  The new       ODSTEXT procedure makes it possible.

PROC ODSTEXT is a SAS 9.4 procedure that allows you to specify text       in the form of paragraphs and nested lists.  You specify paragraphs       with the P statement.  The LIST, ITEM, and END statements are for       defining lists.  If you are familiar with how HTML defines paragraphs       and lists, then these statements should make immediate sense to you.

As the example demonstrates, ODS inline formatting works with       PROC ODSTEXT text.  The code specifies the STYLE inline formatting       function to make the first letter of the text of each list item have a       red foreground color.

Notice that the second P statement in the example has text that breaks       across a line.  The intent is to format the long text in a way that is       readable and maintainable.  After all, this is a SAS program!

Formatting text in this way introduces whitespace characters.  Because       iBooks ignores the extra whitespace, the rendered text is unaffected.        When you are writing your e-book, take advantage of this behavior and       write text that does not wrap, but rather breaks sanely.

Text in EPUB e-books is designed to adapt to the display size of the       device rendering it.  This feature, called        reflowable text, permits you to read EPUB e-books on       mobile devices of varying form factors.  Figure 12 shows how iBooks       displays the e-book  Visit 50 States on a       fourth-generation iPod touch.






[image: A screenshot of iBooks on an iPod displaying the first version of visit_50_states.epub]

Figure 12. Rich Text in iBooks on iPod





You probably do not want to do much reading on an iPod, but you can!

Aside from its unique ability to generate rich text as paragraphs and       lists, PROC ODSTEXT is different from other ODS-integrated procedures       in another way:  the other procedures issue initial page breaks, and       PROC ODSTEXT does not.  Consider this revision to the SAS program that       generates  Visit 50 States:

Show preformatted output
Example 4







Figure 13 shows how iBooks displays the revised e-book.  (Note that the       iBooks font size has been reduced so that all of the text fits on a       single page.)






[image: A screenshot of iBooks displaying the second version of visit_50_states.epub]

Figure 13. More Rich Text in iBooks





The added text does not begin on a new page, because typically you do       not want it to behave that way.  Unless you are starting a new chapter       or section, text should flow in an uninterrupted fashion.  To begin a       new section, you can specify the PROC ODSTEXT statement PAGEBREAK       option.  To learn how to begin a new chapter, keep reading!

The example demonstrates another use of the STYLE inline formatting       function, which is to create a hyperlink.  Tapping the hyperlink       launches Safari at the specified location.  In the example, tapping       the text "Four Corners" takes you to a Wikipedia entry about the       Four Corners region of Colorado, New Mexico, Arizona, and Utah.

Embedding links to external resources enriches the content of your       e-book, but also makes it less portable.  You should not write your       EPUB e-book in such a way that the reader must have an Internet       connection to benefit from reading it.  Make sure that your e-book       is equally readable in the office or at the beach!





START A NEW CHAPTER


Physical books group paragraphs into sections and sections into       chapters.  EPUB e-books can have the same arrangement.  Does it       matter?  Can an e-book reader tell the difference between a section       and a chapter?







From a user perspective, the answer is no–sections and       chapters behave in the same way.  However, from a performance       perspective, there is a difference.  EPUB e-book readers load       e-books into computer memory a chapter at a time.  If you write a       lengthy e-book, organizing its material into chapters will allow       the e-book reader to load it faster, and with a smaller memory       footprint, than if you structure the e-book as a single chapter.

Consider another revision to the example e-book        Visit 50 States.  This revision adds an appendix       chapter that lists attractions seen in visited states.  Figure 14       shows how iBooks displays the new chapter.






[image: A screenshot of iBooks displaying page 2 and 3 of the third version of visit_50_states.epub. Page 2 has 3 lines of text. Page 3 starts a new chapter, 'Appendix', and has 2 paragraphs and a list.]

Figure 14. Appendix Chapter in iBooks





Here is the SAS code that generates the e-book displayed in Figure 14.

Show preformatted output
Example 5







As you see from the code, you start a new chapter with the ODS EPUB       statement NEWCHAPTER option.  The NEWCHAPTER option accepts several       values.  The example demonstrates the value NOW, which should be       self-explanatory.  The other values are:


	PROC:  Start a new chapter for every procedure.

	OUTPUT:  Start a new chapter for every output object.

	PAGE:  Start a new chapter for every explicit page break.

	BYGROUP:  Start a new chapter for every BY group.

	NONE:  Do not start a new chapter.  NONE is the default value for the NEWCHAPTER option.



The other code change deserving comment is the use of the LIST       statement STYLE option to suppress the display of disc (bullet)       markers for the list of visited states.  Every LIST statement list       has a style type, which is governed by the LISTSTYLETYPE attribute.        PROC ODSTEXT supports many list style types, including        decimal,  disc,        square, and  upper-roman.         Disc is the default list style       type.  To suppress the display of list markers, specify a style       type of  none.





TABLE YOUR CONTENTS


Up to this point you have explored how PROC ODSTEXT differs from       other SAS procedures.  In other ways, it is the same.  PROC ODSTEXT,       like other procedures, is represented in the table of contents.        Figure 15 shows how iBooks renders the table of contents for the       latest version of the EPUB e-book        Visit 50 States.









[image: A screenshot of the Table of Contents page in iBooks, displaying the third version of visit_50_states.epub. Its items read 'The Odstext Procedure', 'TextBlock', 'The Odstext Procedure', 'Textblock', ']

Figure 15. Default Table of Contents in iBooks





This table of contents is hopelessly generic and of no use to the       reader.  Worse, it practically screams that the book is the product       of an automated process, and an inferior one at that.  The default       table of contents must go!

Two repeated SAS code modifications will improve the table of contents:


	Specify the ODS statement PROCLABEL option to customize the            procedure entries ("The Odstext Procedure").

	Specify the PROC ODSTEXT CONTENTS option to suppress the            output object entries ("TextBlock").



The modified SAS code follows.

Show preformatted output
Example 6







Figure 16 displays the custom table of contents.






[image: A screenshot of the Table of Contents page in iBooks, displaying the fourth version of visit_50_states.epub. Its items read 'SMART Traveling', 'The Journey's End', 'Appendix']

Figure 16. Custom Table of Contents in iBooks





To further enhance the  Visit 50 States e-book,       you could add       headings to the text sections that correspond to the table of       contents entries.   That would help the reader to know where on the       page to start reading when he selects a section.  The first chapter       of  Visit 50 States has two sections.  The second       (appendix) chapter already has a title, so it does not need       additional context.  How to add section headings is left as an       exercise for the reader.

You might have noticed that the appearance of the cover of        Visit 50 States changed from Figure 15 to Figure 16.        iBooks determines the default cover, and it selects from a variety of       built-in cover designs according to some unknown criteria.        Fortunately, you can customize the cover in iBooks.  You will learn       how to do that, plus more, next!






title;                                                                          
ods escapechar='^';                                                             
ods epub file="visit_50_states.epub" title="Visit 50 States"                    
         options(creator="David W. Kelley");                                    
proc odstext;                                                                   
   p "I learned from SAS employee training that, ideally, a goal should be SMART
:";                                                                             
   list;                                                                        
     item "^{style [foreground=red]S}pecific";                                  
     item "^{style [foreground=red]M}easurable";                                
     item "^{style [foreground=red]A}ttainable";                                
     item "^{style [foreground=red]R}elevant";                                  
     item "^{style [foreground=red]T}ime-bound";                                
   end;                                                                         
   p "I'm confident that our goal of visiting 50 states is measurable,          
      attainable, relevant, and time-bound.  But is it specific?";              
run;                                                                            
ods epub close;                                                                 




title;                                                                          
ods escapechar='^';                                                             
ods epub file="visit_50_states.epub" title="Visit 50 States"                    
         options(creator="David W. Kelley");                                    
proc odstext;                                                                   
   p "I learned from SAS employee training that, ideally, a goal should be SMART
:";                                                                             
   list;                                                                        
     item "^{style [foreground=red]S}pecific";                                  
     item "^{style [foreground=red]M}easurable";                                
     item "^{style [foreground=red]A}ttainable";                                
     item "^{style [foreground=red]R}elevant";                                  
     item "^{style [foreground=red]T}ime-bound";                                
   end;                                                                         
   p "I'm confident that our goal of visiting 50 states is measurable,          
      attainable, relevant, and time-bound.  But is it specific?";              
run;                                                                            
                                                                                
proc odstext;                                                                   
   p "While it's fun to measure our progress towards our goal of visiting all   
      50 states, the greatest enjoyment is in the travel itself.  That is       
      another way of saying that it's never too late to join the ranks of       
      aspiring 50-staters. You get to make your own travel plans, and to        
      formulate your own rules about what qualifies as a state visit.";         
                                                                                
   p "That said, whatever else you do, don't interrupt a through-trip long      
      enough to stand at                                                        
      ^{style [url='http://en.wikipedia.org/wiki/Four_Corners']Four Corners}    
      and then claim that you've visited those four states.  That's plain       
      cheating!";                                                               
run;                                                                            
ods epub close;                                                                 




title;                                                                          
ods escapechar='^';                                                             
ods epub file="visit_50_states.epub" title="Visit 50 States"                    
         options(creator="David W. Kelley");                                    
proc odstext;                                                                   
   p "I learned from SAS employee training that, ideally, a goal should be SMART
:";                                                                             
   list;                                                                        
     item "^{style [foreground=red]S}pecific";                                  
     item "^{style [foreground=red]M}easurable";                                
     item "^{style [foreground=red]A}ttainable";                                
     item "^{style [foreground=red]R}elevant";                                  
     item "^{style [foreground=red]T}ime-bound";                                
   end;                                                                         
   p "I'm confident that our goal of visiting 50 states is measurable,          
      attainable, relevant, and time-bound.  But is it specific?";              
run;                                                                            
                                                                                
proc odstext;                                                                   
   p "While it's fun to measure our progress towards our goal of visiting all   
      50 states, the greatest enjoyment is in the travel itself.  That is       
      another way of saying that it's never too late to join the ranks of       
      aspiring 50-staters. You get to make your own travel plans, and to        
      formulate your own rules about what qualifies as a state visit.";         
                                                                                
   p "That said, whatever else you do, don't interrupt a through-trip long      
      enough to stand at                                                        
      ^{style [url='http://en.wikipedia.org/wiki/Four_Corners']Four Corners}    
      and then claim that you've visited those four states.  That's plain       
      cheating!";                                                               
run;                                                                            
                                                                                
ods epub newchapter=now;                                                        
                                                                                
title "Appendix";                                                               
                                                                                
proc odstext;                                                                   
   p "Here I list one, and only one, memorable attraction from each state       
      we visited.  Limiting each state to one attraction is unreasonable,       
      perhaps even ridiculous.  My flimsy rationale is that no list could       
      adequately cover the subject, so why even try?";                          
   p "With that boulder-sized caveat put forth, here is my list:";              
   list/style={liststyletype="none"};                                           
     item "Arizona:                                                             
           ^{style [url='http://www.americansouthwest.net/arizona/meteor_crater/
index.html']Meteor Crater}";                                                    
     item "Colorado:                                                            
           ^{style [url='http://en.wikipedia.org/wiki/Pikes_Peak']Pikes Peak}"; 
     item "South Dakota:                                                        
           ^{style [url='http://crazyhorsememorial.org/']Crazy Horse Memorial}";
     item "Washington:                                                          
           ^{style [url='http://www.nps.gov/olym/planyourvisit/visiting-hurrican
e-ridge.htm']Hurricane Ridge}";                                                 
   end;                                                                         
run;                                                                            
                                                                                
ods epub close;                                                                 




title;                                                                          
ods escapechar='^';                                                             
ods epub file="visit_50_states.epub" title="Visit 50 States"                    
         options(creator="David W. Kelley");                                    
ods proclabel = "SMART Traveling";                                              
proc odstext contents="";                                                       
   p "I learned from SAS employee training that, ideally, a goal should be SMART
:";                                                                             
   list;                                                                        
     item "^{style [foreground=red]S}pecific";                                  
     item "^{style [foreground=red]M}easurable";                                
     item "^{style [foreground=red]A}ttainable";                                
     item "^{style [foreground=red]R}elevant";                                  
     item "^{style [foreground=red]T}ime-bound";                                
   end;                                                                         
   p "I'm confident that our goal of visiting 50 states is measurable,          
      attainable, relevant, and time-bound.  But is it specific?";              
run;                                                                            
                                                                                
ods proclabel = "The Journey's End";                                            
proc odstext contents="";                                                       
   p "While it's fun to measure our progress towards our goal of visiting all   
      50 states, the greatest enjoyment is in the travel itself.  That is       
      another way of saying that it's never too late to join the ranks of       
      aspiring 50-staters. You get to make your own travel plans, and to        
      formulate your own rules about what qualifies as a state visit.";         
                                                                                
   p "That said, whatever else you do, don't interrupt a through-trip long      
      enough to stand at                                                        
      ^{style [url='http://en.wikipedia.org/wiki/Four_Corners']Four Corners}    
      and then claim that you've visited those four states.  That's plain       
      cheating!";                                                               
run;                                                                            
                                                                                
ods epub newchapter=now;                                                        
                                                                                
ods proclabel = "Appendix";                                                     
title "Appendix";                                                               
                                                                                
proc odstext contents="";                                                       
   p "Here I list one, and only one, memorable attraction from each state       
      we visited.  Limiting each state to one attraction is unreasonable,       
      perhaps even ridiculous.  My flimsy rationale is that no list could       
      adequately cover the subject, so why even try?";                          
   p "With that boulder-sized caveat put forth, here is my list:";              
   list/style={liststyletype="none"};                                           
     item "Arizona:                                                             
           ^{style [url='http://www.americansouthwest.net/arizona/meteor_crater/
index.html']Meteor Crater}";                                                    
     item "Colorado:                                                            
           ^{style [url='http://en.wikipedia.org/wiki/Pikes_Peak']Pikes Peak}"; 
     item "South Dakota:                                                        
           ^{style [url='http://crazyhorsememorial.org/']Crazy Horse Memorial}";
     item "Washington:                                                          
           ^{style [url='http://www.nps.gov/olym/planyourvisit/visiting-hurrican
e-ridge.htm']Hurricane Ridge}";                                                 
   end;                                                                         
run;                                                                            
                                                                                
ods epub close;                                                                 





OPTIMIZE IBOOKS 




You have learned how to create an ODS EPUB e-book for your Apple       mobile device, and how to enrich its content with text arranged       into lists, paragraphs, sections, and chapters.  You know how to       customize the table of contents to facilitate the reader's       navigation in the e-book.  Now the discussion turns to making       the best use of ODS EPUB and iBooks to write e-books that are       visually appealing, thoughtfully constructed, and accessible to       a wide range of readers.





JUDGE A BOOK BY ITS COVER


Ideally your ODS EPUB e-book's cover will catch your reader's eye and       entice her to read the e-book.  By that standard, the iBooks default       cover designs, which are serviceable but visually uninteresting, fall       short.  No doubt you could do better, whether on your own or with       professional assistance.  Fortunately, you can customize the cover       for your e-book.  iBooks and other EPUB e-book readers will display       your custom cover.







To specify a cover for your e-book, use the COVER_IMAGE suboption of       the ODS EPUB statement OPTIONS option.  For example, to specify a       cover for the  Visit 50 States e-book, modify the       initial ODS EPUB statement to be:






[image: images/epubimage3.png]

The COVER_IMAGE value is an image filename or fileref.  The image       type should be JPEG for photographs and PNG for anything else.        For best results in iBooks, Apple recommends that the cover image       be 2400 pixels along the shorter axis.

Figure 17 shows how iBooks displays the custom cover of        Visit 50 States.






[image: A screenshot of the Table of Contents page in iBooks, displaying the fifth version of visit_50_states.epub. There is a custom image for the cover picture.]

Figure 17. Custom Cover in iBooks





The cover image in Figure 17 has dimensions of 2304 pixels by 3072       pixels, as determined by the camera with which the photograph was       taken.  (In case you are wondering, the photograph was taken in the       Rocky Mountain National Park in Colorado.)

If you supply a custom cover for your e-book, then the cover image       should include title and author information.  If your cover is a       photograph, then you will need to edit the photograph.  Photograph       editors range from the simple (Microsoft Paint) to the sophisticated       (Adobe Photoshop).  Microsoft Paint sufficed to edit the cover image       for  Visit 50 States, but it is just an example       e-book, after all.  You can easily do a better job with your       custom cover!





STEP OUT OF LINE


iBooks excels at reflowing text and scaling graphics displays to fit       its page size, which is determined by the display size and       orientation of the reader's mobile device.  It is not nearly as       good at displaying tables that are too tall and/or wide for the page.        iBooks breaks tall tables across pages without repeating the       column headers, which makes it hard to know what is in which column.        Worse, it truncates columns that extend beyond the right margin of       the page.  You can mitigate this undesirable behavior by       double-tapping the table to put iBooks in pan-and-zoom mode.        Nevertheless, SAS makes it easy to generate large tables, and       iBooks needs an acceptable way to display such tables initially.        To its credit, iBooks provides a solution.







The way to handle too-big tables in iBooks is to place them in the       nonlinear section of the e-book.  This is the section of an EPUB       e-book for material that is outside of the linear flow of the       e-book.  By default all ODS EPUB e-book material is linear.        If you want a table to be in the nonlinear section, then you must       specify the NONLINEAR suboption of the ODS EPUB statement OPTIONS       option.

The following SAS code demonstrates the NONLINEAR suboption:

The SAS code specifies NONLINEAR="table" to instruct the ODS EPUB       destination to place tables in the nonlinear section of the e-book        PROC PRINT With 100 Rows. Figure 18 shows how iBooks       displays  PRINT With 100 Rows.






[image: A screenshot of iBooks, displaying on its one page, Non-linear Table, heading level 1, Data Set SASHELP.PRDSAL2, Show table, link.]

Figure 18. Nonlinear Table in iBooks





Instead of displaying the table, iBooks displays an icon for the       table.  Selecting the icon puts iBooks in pan-and-zoom mode on the       table–all 100 intact rows of it.  Much better!  With       the NONLINEAR suboption, you can generate an EPUB e-book that iBooks       will handle gracefully, no matter the size of any table in the       e-book.





BE ACCESSIBLE


Apple iPads, iPhones, and iPods contain built-in accessibility      features that allow users of all abilities to enjoy digital content.       These features include the ability to customize font size and color       schemes, magnification, and the VoiceOver screen reader, which       allows totally blind users to use iOS devices. iBooks is fully       compatible with these accessibility features. In addition, the       EPUB specification has been specifically designed to ensure that       e-books are accessible to users with disabilities.







ODS EPUB leverages this infrastructure. It will do most of the work       required for your e-book to be accessible to a broad audience.       However, there are a few things that you can do to ensure everyone       can enjoy your e-book.





CREATE MEANINGFUL LINK TEXT


Ensure that the purpose of the links in your e-book can be determined       from only the link text when they are viewed out of context. This       will allow users with visual impairments to quickly scan the links       within your e-book using the VoiceOver screen reader.







For example, the link text in the first sentence below does not       convey the purpose of the link when viewed out of context. However,       the link text in the second sentence does.

Click here to learn how to       customize the cover of your e-book.

See JUDGE A BOOK BY ITS COVER to       learn how to customize the cover of your e-book.





DO NOT USE COLOR ALONE TO CONVEY MEANING

Ensure that users with color-blindness can perceive the graphs and       charts in your e-book by using an additional characteristic to       convey meaning. For example, the chart below uses both color and      shape to identify the gender of the student represented by each data       point. The colors chosen are also not only different hues (blue and       green) but also different color brightness levels (light and dark). A       reader with blue-green color-blindness will still be able to        distinguish between the colors, as would a reader receiving a       black-and-white print-out of this chart.








[image: A scatterplot titled 'Students' with 19 data points, plotting Height (in), Weight (lbs), and Gender. Female students are represented with light blue squares, and male students are represented with dar]

Figure 19. A Scatter Plot Using Color and Shape





This scatter plot was created by the following SAS code:

IDENTIFY COLUMN AND ROW HEADERS IN TABLES


Ensure that users with visual impairments can understand the tables       in your e-book by identifying the column and row headers when you       create the tables. This will cause SAS to mark up table cells that       contain headers so that VoiceOver can describe the column and row       header information for each cell in the table.









[image: Table output that contains column headers and row headers.]

Figure 20. PROC PRINT Output with Row Headers





The table above was created using the following SAS code:






[image: images/epubimage3.png]

STYLE TABLES


The size of the text within tables can be very small by default,       which makes it difficult for users to read the table. Also, the       height and width of cells within tables can be very small by default,       which makes it difficult for totally blind users to explore tables       using VoiceOver. Ensure that the text and cells in tables are large       enough to meet the needs of all users.

[image: A Table with 2 columns and 20 rows. The column headers are 'Name' and 'Sex.' The table is sorted by 'Sex'.]

Figure 21. PROC PRINT Output with the Previous Template Applied






PROVIDE TEXT ALTERNATIVES FOR IMAGES


Users with visual impairments might not be able to see the charts,       graphs, and other images in your e-book. Be sure to include a short       text alternative that summarizes the purpose of each image in your       e-book. Do not include "Image of ...", "Picture of ..." or similar       phrases in the text alternative. It is not necessary to include       those phrases because VoiceOver communicates the type of each object       to the user. For example, the text alternative specified by the SAS       code below is perfectly adequate.

can also provide custom text alternatives for your SAS output.

[image: Bar chart of students by gender]

Figure 22. SAS Bar Chart with Text Alternative (Not Shown Here)





When totally blind users encounter the chart in Figure , they will       hear "Bar chart of students by gender".

For complex charts and graphs, you should also include a long       description of the image. You can do this by including a verbal       description inline within the text of the e-book or by providing       a link to a NONLINEAR section that contains a table of the data       represented in the image.

The following SAS code creates a NONLINEAR section that represents       the data in the bar chart above in table format.

ods epub file="visit_50_states.epub" title="Visit 50 States"                    
         options(creator="David W. Kelley" cover_image="images/cover.jpg");     




ods epub file="print100rows.epub" title="PROC PRINT With 100 Rows"              
         options(creator="David W. Kelley" nonlinear="table");                  
title 'Non-linear Table';                                                       
proc print data=sashelp.prdsal2(obs=100);                                       
run;                                                                            
ods epub close;                                                                 
title;                                                                          




gsymbol1 value=squarefilled color=cyan height=3;                                
symbol2 value=trianglefilled color=green height=3;                              
axis1 label=(angle=90 height=2 "Height (in)" ) order=(50 to 80 by 5);           
axis2 label=(height=2 "Weight (lbs)");                                          
legend1 label=none value=("Female" "Male");                                     
title "Students";                                                               
                                                                                
proc gplot data=sashelp.class;                                                  
	plot height*weight=sex / legend=legend1 vaxis=axis1 haxis=axis2;               
run;                                                                            




proc print data=sashelp.class(where=(age=12)) noobs;                            
	var name / style=rowheader;                                                    
	var age weight height;                                                         
run;                                                                            




proc template;                                                                  
   define style styles.docStyle;                                                
   parent = styles.daisy;                                                       
                                                                                
	class header /                                                                 
		fontweight = bold                                                             
		width=90px;                                                                   
	class rowheader /                                                              
		fontweight = bold                                                             
		fontsize = 15pt                                                               
		height=20px                                                                   
		width=100px;                                                                  
	class data /                                                                   
		fontsize=13pt;                                                                
   end;                                                                         
run;                                                                            




proc odstext contents="";                                                       
	p "There are some dangerous animals in our forests. It is best to avoid them.";
	p "^{image &path\bear_20070606.png?alt_desc=Grizzly Bear}"/style=image;        
run;                                                                            




proc gchart data=sashelp.class;                                                 
	vbar sex / description="Bar chart of students by gender ";                     
run;                                                                            




proc template;                                                                  
   define style styles.docStyle;                                                
   parent = styles.daisy;                                                       
                                                                                
	class header /                                                                 
		fontweight = bold                                                             
		fontsize = 15pt                                                               
		width=90px;                                                                   
	class rowheader /                                                              
		fontweight = bold                                                             
		fontsize = 15pt                                                               
		height=20px                                                                   
		width=100px;                                                                  
	class data /                                                                   
		fontsize=13pt;                                                                
   end;                                                                         
run;                                                                            
                                                                                
ods epub file="students.epub" title="Students" options(nonlinear="table" content
s="off") style=docStyle;                                                        
                                                                                
title "Girls and Boys";                                                         
axis1 label=("Sex") order=("F" "M") length=80;                                  
axis2 label=(angle=90 "Number of Students") order=(0 to 10 by 1);               
proc gchart data=sashelp.class;                                                 
	vbar sex / description="Bar chart of students by gender"                       
			maxis=axis1 raxis=axis2 width=30;                                            
run;                                                                            
                                                                                
proc sort out=temp data=sashelp.class(keep=name sex); by sex; run;              
                                                                                
proc print data=temp noobs;                                                     
	var name / style=rowheader;                                                    
	var sex;                                                                       
run;                                                                            
ods epub close;                                                                 





CONCLUSION 




Apple had a famous advertising campaign with the slogan        Think Different.  Hopefully after reading this       paper or viewing the presentation, you will think different       about how you will use ODS in SAS 9.4.  Move beyond traditional       office-based reporting and deliver your SAS output as EPUB e-books       on mobile devices.  Write a novel with SAS and self-publish it.        Easily share your e-books with customers and friends, including       visually impaired readers.  Now that you know how to do it, get       to it!

